
				1	Einführung in Processing - Visual Java für Einsteiger

				Shortcut - Gliederung

				Kapitel 1: Einführung ins Processing, Nutzen der 2-D-Rendering-Engine

				Kapitel 2: Nutzen der 3-D-Rendering-Engine mit Kamerafahrten

				Kapitel 3: Computer Vision und Augmented Reality mit Processing

				Kapitel 4: Professionelle Datenvisualisierung mit Java
				Java hat sich in vielen Bereichen der Informatik unverzichtbar gemacht – was auch gut so ist. Im Fokus dieses shortcut steht allerdings nicht Java, sondern eine andere, auf Java basierende Open-Source-Programmiersprache samt Entwicklungsumgebung: Processing. In deren Umfeld ist es ein Kinderspiel, MS Kinect oder Playstation-Controller anzuschließen, Computerbilderkennung zu betreiben, ästhetische – auch ungewöhnliche – Visualisierungen zu erstellen oder einfach nur unterhaltsame und interaktive Anwendungen zu bauen. Im ersten Kapitel des shortcuts möchten wir die Basics abdecken, bevor wir im zweiten auf die einzigartigen Möglichkeiten eingehen.

				Processing ist eine Programmiersprache samt Entwicklungsumgebung, die seit ihrer Erfindung am MIT 2001 kontinuierlich weiterentwickelt wird. Ursprünglich wurde die Sprache erschaffen, um Programmieranfängern eine visuelle Lernumgebung zu bieten. Mittlerweile kann man aber auch professionelle und abgeschlossene Anwendungen erstellen. Aufgrund der knappen Syntax und der visualisierungslastigen APIs bietet Processing sich bei der Erstellung von Grafiken, Animationen und natürlich interaktiven Anwendungen an.

				[image: Abb1_1.png]

				Abbildung 1.1: Processing IDE im JavaScript-Modus

				Da Processing auf Java aufbaut, ist die Sprache selbst sowohl objektorientiert als auch stark typisiert. Durch eine vereinfachte Syntax und reichhaltige Kontextobjekte ist der Quelltext sehr lesbar und knapp gehalten im Vergleich zu regulären Java-Anwendungen. Schauen wir uns das erste Programm – in Processing auch Sketch genannt – an (Listing 1.1).

				void setup() {
  frameRate(24);
  smooth();
  size(600, 600);
}
void draw() {
  //Alpha Box
  fill(255, 20);
  noStroke();
  rect(0,0,600,600);
  //Dreieck
  fill(255);
  stroke(0); 
  translate(300, 300);
  rotate(radians(frameCount*2));
  triangle(-150, -86, 0, 174, 150, -86);
}

				Listing 1.1

				Als Java-Entwickler erkennt man die Syntax wieder, und man kann sofort zwei Methodenimplementierungen erkennen, vermisst allerdings die Klassendefinition. Dies geschieht implizit und leitet sich von der Basisklasse PApplet ab. Diese Basisklasse sieht nicht nur die Hauptmethoden setup() und draw() vor, sondern stellt eine ganze Reihe von Methoden zur grafischen Ausgabe, für verschiedene Eingabemechanismen, Farbberechnung, mathematische Funktionen und vieles mehr zur Verfügung.

				In der setup()-Methode wird die Bildrate auf 24 Wiederholungen pro Sekunde gesetzt, das Anti-Aliasing für die 2-D-Ausgabe aktiviert und eine Ausgabefenstergröße von 600 mal 600 Pixel erstellt. Setup() wird einmalig vor der ersten Ausgabe durch Processing aufgerufen und dient daher der Initialisierung des Sketch.

				Die draw()-Methode wird hingegen für jede Aktualisierung des Bilds aufgerufen – in unserem Beispiel 24-mal pro Sekunde. Beim Neuzeichnen des Bildschirms wird die vorherige Ausgabe nicht gelöscht. Das nutzen wir, um einen Fade-Effekt zu generieren, bei dem sich vorherige Zeichenoperationen graduell ausblenden lassen. Wir legen hierzu ein semitransparentes weißes Rechteck über das gesamte Fenster. Die fill()-Anweisung definiert die Farbe und den Alpha-Wert, noStroke() verzichtet auf einen gezeichneten Rand, und rect() zeichnet letztendlich das Quadrat.

				Jetzt wollen wir das Dreieck rotieren lassen. Zunächst müssen wir die Zeichenoperationen für die Füllung und Umrandungen wieder auf volles Weiß bzw. Schwarz mit fill() und stroke() zurücksetzen. Das Dreieck soll um den Mittelpunkt des Umkreises rotiert werden. Da Processing alle Rotationsoperationen um den Ursprung (0|0) ausführt, muss das Dreieck so gezeichnet werden, dass der Mittelpunkt im Koordinatenursprung liegt. Das geschieht mit der Methode triangle(). Damit das Dreieck über die obere linke Ecke rotiert wird, verschieben wir das Koordinatensystem um 300 Pixel nach rechts und unten mit der translate()-Anweisung. Zu guter Letzt wird aus der Anzahl der bisherigen Wiederholungen der Drehwinkel bestimmt, und per rotate() wird die Rotation schlussendlich auch ausgeführt.

				Abbildung 1.2: Grafische Ausgabe der Dreiecksrotation

				Die Anwendung kann direkt aus der IDE gestartet werden, die im Hintergrund aus dem Processing Code vollwertigen Java Sourcecode für ein Applet erstellt, kompiliert und das Applet anschließend startet. In der neuesten Version ist es möglich, den Processing-Code per HTML5 und Canvas direkt im Browser auszuführen oder als Android-Anwendung im Emulator zu starten. Diese so genannten Modes (Java, Android und JavaScript) können direkt in der IDE ausgewählt werden. Wenn man mit der Entwicklung der Anwendung fertig ist, kann man den Sketch für das entsprechende Modul exportieren – damit lässt sie sich ohne IDE direkt auf dem Desktop, Browser bzw. Android-Gerät starten.

				Ansprechende Visualisierungen mal anders

				Ein weiteres Beispiel, wie man mit sehr wenig Code spannende Ausgaben erzeugen kann, ist die Erzeugung grafischer Auswertungen von regionalen Daten. Die folgende Anwendung (Listing 1.2) visualisiert die Fitch Ratings der EU-Nationen. Die grünen Töne stellen gute, die roten Töne weniger gute Ratings dar.

				import java.util.Map;
import java.util.Iterator;

PShape europe;
String[] euCountries = {"be", "bg", "dk", "de", "ee", "fi", "fr", "gr", "ie", "it", "lv", "lt", "lu", "mt", "nl", "at", "pl", "pt", "ro", "se", "sk", "si", "es", "cz", "hu", "uk", "cy", "hr" };
String[] fitchRatings = {"AA", "BBB-", "AAA", "AAA", "A+", "AAA", "AAA", "B-", "BBB+", "A-", "BBB-", "BBB", "AAA", "A+", "AAA", "AAA", "A-", "BB+", "BBB-", "AAA", "A+", "A", "BBB", "A+", "BBB-", "AAA", "B", "BBB-" };
String[] ratingList = {"AAA", "AAA-", "AA+", "AA", "AA-","A+", "A", "A-","BBB+", "BBB", "BBB-", "BB+", "BB", "BB-","B+", "B", "B-","CCC+", "CCC", "CCC-", "CC+", "CC", "CC-","C+", "C", "C-"};

HashMap<String, String> fitchRatingsEurope = new HashMap();
HashMap<String, Integer> ratingColors = new HashMap();


void setup(){
  for(int i = 0; i < euCountries.length; ++i){
    fitchRatingsEurope.put(euCountries[i], fitchRatings[i]);
  }
  for(int i = 0; i < ratingList.length; ++i){
    ratingColors.put(ratingList[i], color(i*10, 255-(i*10), 0));
  }
  size(800, 600);
  background(255);
  europe = loadShape("Map.svg");
}

void draw(){
  scale(0.075);
  europe.disableStyle();
  fill(color(128,128,128));
  shape(europe, 0, 0);
  drawCountries(fitchRatingsEurope, ratingColors);
}

				Listing 1.2 

				Die Hashmaps werden in der setup()-Methode aus den String-Arrays befüllt. Die HashMap fitchRatingsEurope enthält das ISO-Kürzel der Länder als Schlüssel und die Ratings als Werte. Die ratingColors Map hat wieder die Länderkürzel als Schlüssel, enthält aber berechnete Farbcodes als Werte. Man hätte dies alternativ direkt aus z.B. einer XML-Datei (loadXML()) laden können, aber aus Anschauungszwecken wurden die Daten direkt inline im Quellcode abgelegt. In der setup()-Methode wird außerdem eine SVG-Grafik geladen. Die Grafik enthält die Umrisse aller europäischen Staaten. Mit Inkscape wurde sichergestellt, dass die Umrisse das ISO-Kürzel des jeweiligen Landes als ID tragen.

				In der draw()-Methode wird die Ausgabe verkleinert; die CSS-Eigenschaften der Karte wurden deaktiviert und ein Grauton wurde als Standardfüllfarbe definiert. Mit diesen Angaben werden die Umrisse gezeichnet. In der Hilfsmethode drawCountries() wird für jedes Land der EU der entsprechende Farbcode für das Ranking ermittelt und das Land mit der entsprechenden Füllung gezeichnet. Da die Umrisse in der SVG-Datei zuvor mit den korrekten IDs versehen worden sind, kann man europe.getChild() sehr intuitiv nutzen.

				Abbildung 1.3: Eingefärbte Europakarte

				Koordinatensystem, Zeichnen und Transformationen

				Das zweidimensionale Koordinatensystem hat seinen Ursprung in der oberen linken Ecke und erstreckt sich von dort aus auf das gesamte Fenster, welches mit der size(width, heigh) erstellt wurde. Neben einfachen geometrischen Basisformen:

				
						Ellipsensegmenten: arc()

						Ellipsen: ellipse()

						Linien: line()

						Punkten: point()

						Quadraten: quad()

						Rechtecken: rect()

						Dreiecken: triangle()

				

				kann man auch beliebig komplexe Formen auf den Bildschirm ausgeben. Diese Formen lassen sich als:

				
						Kurven mit den bezier()- oder den curve()-Funktionen erstellen.

						Polygone mit den vertex()-Funktionen erstellen.

				

				Das Zeichnen in Processing kann man sich sehr bildhaft vorstellen. Man kann sich sein Zeicheninstrument festlegen, und alle folgenden Zeichenoperationen erfolgen mit dem Zeicheninstrument, bis man es wieder ändert. Zum Festlegen der Zeichenoperationen kann man die Funktionen: 

				
						strokeCap(), strokeJoin(), strokeWeight(), stroke() und noStroke() für die Definition der Linien

						fill(), noFill() für die Definition der Füllung 

				

				nutzen (Listing 1.3).

				void setup(){
  size(300,300);
  smooth();
}

void draw(){
  noStroke();
  fill(color(255,0,0));
  
  beginShape();
  vertex(250, 110); 
  vertex(250, 290); 
  vertex(50, 290); 
  vertex(50, 110); 
  endShape();
  
  fill(color(0,255,0));
  bezier(50, 110, 50, 10, 250, 110, 250, 110);
}

				Listing 1.3

				Erwähnenswert sind auch die Translation-Funktionen. Damit kann man das Koordinatensystem, auf dem die Zeichenausgaben erfolgen, manipulieren, sodass die Ausgaben gedreht, gestaucht, gedehnt o.ä. werden (siehe Kasten: „Translation-Funktionen“).

				Translation-Funktionen

				Echte IDEs und los geht’s

				Es sollte nicht unerwähnt bleiben, dass man auch ohne die sehr einfach gehaltene IDE arbeiten kann. Durch die Einbindung der Processing JARs kann man auch ohne Weiteres die Anwendungen in Eclipse schreiben. Hierzu gibt es eine sehr gute Anleitung auf der Processing-Homepage [1]. Man muss lediglich die Startklasse von der erwähnten PApplet-Klasse ableiten, und schon kann es losgehen. Natürlich sind mit Eclipse etc. nur Java-Module nutzbar. Benötigen Sie JavaScript- oder Android-Unterstützung, sollten Sie weiterhin die Processing IDE nutzen.

				Falls Sie nun Interesse haben, schauen Sie auf der Downloadseite [2] vorbei und laden Sie das Paket herunter. Packen Sie das Archiv aus, starten Sie die IDE – und in wenigen Minuten läuft Ihre erste Anwendung. Happy Sketching.

				Links & Literatur

				
						http://processing.org/learning/eclipse/

						http://processing.org/download/

				

			

		

	
		
			
				2	Processing: Besonderheiten und Möglichkeiten

				3-D-Programmierung

				Processing ist eine auf Java basierende Open-Source-Programmiersprache samt Entwicklungsumgebung. Im ersten Kapitel dieses shortcut wurden Grundlagen und Ursprünge dieses Visual-Java-Tools erläutert und ein paar Fingerübungen mit der 2-D-Engine gemacht. Nun möchten wir einem aktuellen Trend in der Filmindustrie folgen und überprüfen, was Processing im 3-D-Bereich zu bieten hat.

				Bisher haben wir immer den Standard 2-D-Rendermode verwendet. Dieser wird implizit beim Initialisieren des Bildschirms gesetzt:

				size(800, 600)

				Daher ist der Schritt bei der Ausgabe von 3-D die explizite Benennung des Modus. Hierzu gibt es eine Reihe von Modi (Tabelle 2.1).

				
					
						
								
								Modus

							
								
								Beschreibung

							
						

						
								
								P2D

							
								
								Default-2-D-Renderer (nutzt intern OpenGL)

							
						

						
								
								P3D

							
								
								Default-3-D-Renderer (nutzt intern OpenGL)

							
						

						
								
								OPENGL

							
								
								Bietet direkten Zugriff auf das JOGL API

							
						

						
								
								PDF

							
								
								Vector 2-D-Shaped in PDF-Dateien

							
						

					
				

				Tabelle 2.1: Render-Methoden

				Dieses Kapitel wird sich ausschließlich mit P3D beschäftigen, da hier die Vorteile von Processing unterstrichen werden sollen – OpenGL bietet lediglich direkten Zugriff auf den Java-OpenGL-Wrapper an. Außerdem gibt es keine Garantien, dass der Modus auch weiterhin durch Processing unterstützt wird. So wurde zum Beispiel schon jetzt die Möglichkeit des Java-Applet-Exports für OpenGL eingeschränkt.

				Der 3-D-Raum leitet sich direkt vom 2-D-Raum ab. Vom Ursprung nach unten zeigt die Y-Achse, die X-Achse verläuft vom Ursprung nach rechts, und die Z-Achse kommt aus dem Ursprung auf den Betrachter zu.

				Die Kamera schaut auf die Mitte des Bildschirms mit z = 0 und hat ihren Ursprung ebenfalls in der Mitte des Bildschirms (Abb. 2.1), besitzt aber einen z-Wert, der ungefähr der Bildschirmhöhe entspricht (die genaue Formel lautet: (Höhe/2)/(tan π/6)).

				 [image: Abb2_1.png]

				Abbildung 2.1: Dimensionen im 3-D-Raum, der Ursprung liegt in O, die Kamera liegt auf dem Punkt C und zeigt auf F

				Genug Theorie, wir wollen anfangen, eine – zugegebenermaßen primitive, aber nachvollziehbare – Darstellung einer beliebigen Großstadt mit Skyline interaktiv darzustellen.

				Aufbau der Szene

				Zunächst benötigen wir einen Himmel, den wir uns mit drawSky() generieren. Hier finden wir eine neue Methode mit dem Namen sphere() zum Zeichnen einer Kugel. Im ersten Kapitel haben wir bereits das 2-D-Gegenstück ellipse() besprochen. totallength ist die Kantenlänge unserer Stadt. Hier benutzen wir sie, um eine Kugel um die gesamte Stadt zu zeichnen. fill() und stroke() sind ebenfalls schon bekannt und dienen dazu, den Himmel blau zu zeichnen. Um den Boden bzw. die Straße zu zeichnen, nutzen wir drawGround(). Auch hier sind die Operationen pushMatrix(), translate() und popMatrix() schon bekannt. Neu ist die box()-Methode, mit der man Quader zeichnen kann. In der Methode zeichnen wir einen flachen Quader mit einer Höhe von 0 (Abb. 2.2).

				Abbildung 2.2: Die Himmelszene (blau), um eine Bodenplatte erweitert

				Die dreidimensionalen Körper können mit fill()farblich gefüllt werden. Um die Flächen mit Bitmaps, also mit Bildern, zu überziehen, muss man in der Regel auf die schon vorgestellten Vertexes zurückgreifen [1]. Im Rahmen des Kapitels möchten wir aber eine externe Bibliothek einbinden, um Körper mit Texturen zu versehen.

				Fremde Bibliotheken einbinden

				Bibliotheken lassen sich sehr einfach einbinden, indem man in der Processing-IDE unter dem Menüpunkt Sketch | Import Library eine der schon vorinstallierten Bibliotheken auswählt. Diese können Tabelle 2.2 entnommen werden.

				
					
						
								
								Standardbibliothek

							
								
								Beschreibung

							
						

						
								
								dxf

							
								
								Export von Modellen in das DXF-Format

							
						

						
								
								lwjgl

							
								
								Unterstützt bei der Erstellung von Spielen

							
						

						
								
								minim

							
								
								Integriert Audiofunktionen

							
						

						
								
								net

							
								
								Zugriff auf diverse Netzwerkroutinen

							
						

						
								
								pdf

							
								
								Export der Ausgabe in das PDF-Format

							
						

						
								
								serial

							
								
								Zugriff auf die virtuellen Serial Ports via USB

							
						

						
								
								video

							
								
								Erlaubt das Abspielen von Videos, Lesen der Webcam etc. 

							
						

					
				

				Tabelle 2.2: Standardbibliotheken

				Neue Bibliotheken lassen sich unter Sketch | Import Library | Add Library... suchen und installieren. Nachdem man nach Shapes3D gesucht, die entsprechende Bibliothek installiert und eingebunden hat, wird man feststellen, dass die entsprechenden Import-Statements am Anfang der Sketch eingebunden werden. Ferner ist es möglich, lauffertige Beispiele für die jeweiligen Bibliotheken anzeigen zu lassen. Hierzu kann man unter File | Examples die entsprechenden Sketches öffnen und ausprobieren. Bibliotheken lassen sich auch per Hand in das Processing/libraries-Verzeichnis im User Home kopieren. 

				Tabelle 2.3: Gebäude werden nun hinzugefügt, haben aber noch keine Textur

				Nach der Vorarbeit kann die initBlocks()-Methode geschrieben werden. Hierzu wird für jeden Block im Raster, analog zu einer amerikanischen Stadt (Abb. 2.3), eine Box (eine der Shapes3D-Klassen) instanziiert. Die Grundfläche entspricht der des Blocks, und die Höhe wird zufällig aus einer Zahl zwischen 120 und 250 gesetzt. Auf den Seiten und der Decke der Box projizieren wir jeweils eine zufällig ausgewählte Textur und speichern das Ergebnis in einem zweidimensionalen Array (Abb. 2.4). Die erste Dimension des Arrays gibt die horizontale Position und die zweite Dimension die vertikale Position im Raster wieder.

				Tabelle 2.4: Gebäude haben nun auch Texturen

				In dem Beispiel haben wir nur eine Box benutzt. Es gibt aber noch weitere Formen, wie etwa dreidimensionale Ellipsen, Spiralen, Röhren, Kegel und vieles mehr. Falls man einen objektorientierten Zugriff bevorzugt oder viel mit Texturen arbeitet, sollte man Shapes3D definitiv anschauen [2].

				Lights, Camera, Action!

				Wir haben nun eine ganz nette Szene an Hochhäusern zusammengestellt. Es fehlt nur noch die Kamera, um das alles aufzunehmen. Processing stellt hier einige Methoden zur Verfügung, um die Kamera zu steuern (Tabelle 2.3).

				
					
						
								
								Methode

							
								
								Beschreibung

							
						

						
								
								camera()

							
								
								Angabe der Position, Blickpunkt und Ausrichtung der Kamera

							
						

						
								
								ortho()

							
								
								Rendern mit einer orthographischen Perspektive

							
						

						
								
								perspective()

							
								
								„normale“ Projektion

							
						

						
								
								frustrum()

							
								
								weitere Perspektivendarstellung, Quadrate werden als Pyramidenstümpfe gerendert 

							
						

						
								
								printCamera()

							
								
								Ausgabe von Debug-Informationen für die Kamera

							
						

						
								
								printProjection()

							
								
								Ausgabe von Perspektiveninformationen

							
						

					
				

				Tabelle 2.5: Methoden für Kamera- und Perspektivensteuerung

				Man kann mit den Funktionen beginCamera() und endCamera() eine Klammer über Transformationen bilden, um die Kamera zu drehen oder ähnliches. Allerdings ist das alles etwas mühselig, sodass wir für schöne Kamerafahrten eine zweite Bibliothek einbinden werden: OCD – Obsessive Camera Direction. Nach dem Suchen und dem Einbinden der Bibliothek kann man sich eine Kamera instanziieren. Auch hier (im Code in der setup()-Methode) gibt man Position, Blickpunkt und Ausrichtung an und kann das Objekt später manipulieren. Da wir die Kameraausrichtung und -position ändern werden, müssen wir die Kameraeinstellungen regelmäßig an die Render-Engine melden, was mittels camera.feed() in der lightsAndCamera()-Methode geschieht.

				Die Kamera werden wir mit der Maus verschieben. Horizontale Bewegungen drehen die Kamera um die Mitte der Stadt, während mit vertikalen Bewegungen die Kamera hinein- und hinausfahren wird. Hierzu nutzen wir eine der Input-Methoden von Processing – in der Java-Welt wäre das ein Event Handler. Die in Tabelle 2.4 aufgelisteten Methoden stehen zur Verfügung.

				
					
						
								
								Methode

							
								
								Beschreibung

							
						

						
								
								mouseMoved()

							
								
								Maus bewegt sich über den Ausgabebildschirm

							
						

						
								
								mouseClicked()

							
								
								Eine der Maustasten wurde gedrückt und losgelassen

							
						

						
								
								mouseDragged ()

							
								
								Die Maus wird mit gedrückter Taste bewegt

							
						

						
								
								mousePressed ()

							
								
								Eine der Maustasten wurde gedrückt

							
						

						
								
								mouseReleased ()

							
								
								Eine der Maustasten wurde losgelassen

							
						

						
								
								keyPressed ()

							
								
								Auf der Tastatur wurde eine Taste gedrückt

							
						

						
								
								keyReleased()

							
								
								Eine der Tasten auf der Tastatur wurde losgelassen

							
						

						
								
								keyTyped

							
								
								Es erfolgte eine Eingabe auf der Tastatur

							
						

					
				

				Tabelle 2.4: Event-Handler-Methoden für Tastatur- und Mauseingaben

				Innerhalb der mouseMoved()-Methode können wir nun mit mouseX und mouseY die aktuelle Mausposition abrufen. Mit pmouseX und pmouseY kann uns Processing auch mitteilen, wo sich die Maus im vorherigen Frame (also zum Zeitpunkt der vorherigen Ausgabe) befand. Mit der Bewegung der Maus zwischen den Frames berechnen wir einfach die Parameter für die jeweiligen Kameramethoden. 

				Die OCD-Kamera verhält sich viel mehr wie eine echte Filmkamera als die originale Processing-Kamera. Viele der Bewegungen eines professionellen Kameramanns sind abrufbar. So kann man die Kamera mit truck(), boom() und dolly() linear verschieben, mit tilt(), pan() und roll() um diverse Achsen drehen oder auch mit arc() und circle() Kreisbewegungen ausführen. Auf der Homepage der Bibliothek [3] sind alle Methoden gut dokumentiert und mit Beispielen versehen.

				Die schönsten Szenen und tollsten Kamerafahrten sind nutzlos, wenn es kein Licht gibt. Ohne weitere Einstellungen wird Processing ein ambientes weißes Licht zur Verfügung stellen. Das Licht hat weder Quelle noch Richtung, sodass alle Flächen gleich ausgeleuchtet und keine Schatten geworfen werden. Damit kann man zwar sehen, wie die Szene aufgebaut ist. Aber spannend ist das nicht.

				In der lightsAndCamera()-Methode wird das ambiente Licht auf einen Grauwert eingestellt, um sehr dunkle und unnatürliche Schatten zu reduzieren. Mit lightFalloff() wird für alle nachfolgend definierten Lichtquellen definiert, wie das Licht über die Entfernung nachlässt. Die Methode akzeptiert drei Parameter für einen konstanten, linearen und quadratischen Abfall. Anschließend gibt es eine punktförmige Lichtquelle, die radial vom Mittelpunkt ausstrahlt. Diese wird mit dem Aufruf pointLight() definiert. Die ersten Parameter bestimmen die Farbe und die letzten Parameter die Koordinaten der Lichtquelle (Abb. 2.5).

				 [image: Abb2_5.png]

				Abbildung 2.5: Finale Szenen mit neuen Lichtquellen und der Kugel zur Kennzeichnung der Lichtquelle

				Eine kleine Auflistung der möglichen Lichtquellen und der entsprechenden Methoden finden Sie in Tabelle 2.5.

				
					
						
								
								Methode

							
								
								Beschreibung

							
						

						
								
								ambientLight ()

							
								
								Licht ohne Richtung und Quelle

							
						

						
								
								directionalLight ()

							
								
								Licht mit spezifischer Richtung, aber ohne Quelle

							
						

						
								
								pointLight ()

							
								
								Radiale Lichtquelle mit spezifischem Ursprung

							
						

						
								
								spotLight ()

							
								
								Licht mit spezifischer Richtung und Quelle

							
						

					
				

				Tabelle 2.5: Methoden zur Gestaltung der Lichtquellen

				

				Damit man auch als Laie erahnen kann, wo die Lichtquelle steht, legen wir unter die Lichtquelle eine weiße Kugel. In diesem Beispiel ausgelassen sind die Möglichkeiten, die Reflektionen der Lichtquellen an den Objekten einzustellen. Man kann damit zum Beispiel bei Stillleben tolle Effekte erzielen. Mit lightSpecular() kann man die Farbe des Lichts einstellen, das in eine konkrete Richtung an einem Objekt reflektiert wird. Die diffusen Reflektionen des Lichts werden weiterhin über die Lichtfarbe eingestellt. Mit den Methoden ambient(), emissive(), shininess() und specular() kann man die optischen Materialeigenschaften einstellen. Beispiele findet man direkt in der Processing IDE unter File | Examples ... und dort unter Basics | Lighting.

				Zusammenfassung

				Leider habe ich einen kleinen Wermutstropfen, der dem aufmerksamen Leser sicherlich aufgefallen ist. Normalerweise kann man mit der Methode loadImage() Bilder ohne Pfadangabe laden, wenn diese zuvor mit Sketch | Add File installiert wurden. Leider hat die aktuellste Processing-Version (2.0b9) auf OSX einen Bug, sodass ich den kompletten Pfad angeben musste. Daher die Konstruktion mit der Methode ldPrefixImg und der Konstanten PREFIX.

				Auch mit dem Workaround konnten wir mit weniger als 100 Zeilen Code, zwei neuen Bibliotheken und ein paar Texturen eine komplette Szene erstellen und ein paar Einblicke in die 3-D-Fähigkeiten von Processing gewinnen (Listing 2.1). Im nächsten Kapitel möchten wir noch tiefer in 3-D einsteigen und eine Augmented-Reality-Anwendung erstellen. Bis dahin kann ich Ihnen nur nochmals die Processing-Reference-Seite [4] ans Herz legen, da sind alle erwähnten Methoden nochmals im Detail erklärt. 

				import shapes3d.utils.*;
import shapes3d.*;
import damkjer.ocd.*;

int blockLength = 100;
int streetWidth = 20;

int blocksTotal = 20;
int totalLength  =  blockLength * blocksTotal + streetWidth * (blocksTotal - 1);
Box[][] blocks = new Box[blocksTotal][blocksTotal]; 
private static final String PREFIX = "/Users/stefansiprell/Pictures/";
PImage[] sides = new PImage[]{ldPrefixImg("/side01.jpeg"), ldPrefixImg("side02.jpeg"),ldPrefixImg("side03.jpeg"),ldPrefixImg("side04.jpeg")};
PImage[] tops = new PImage[]{ldPrefixImg("/top01.jpeg"), ldPrefixImg("top02.jpeg"),ldPrefixImg("top03.jpeg"),ldPrefixImg("top04.jpeg")};
Camera camera;

PImage ldPrefixImg(String name){
  return loadImage(PREFIX+name);}

void setup(){
  size(800, 600, P3D);
  initBlocks();
  camera = new Camera(this, 0, -100, -300, 0, 0, 0, 0, -1, 0); }

void initBlocks(){
 for(int horizontal = 0; horizontal < blocksTotal; horizontal ++){
   for(int vertical = 0; vertical < blocksTotal; vertical ++){
         blocks[horizontal][vertical] = new Box(this, blockLength, random(120, 250), blockLength);
         blocks[horizontal][vertical].setTexture(sides[int(random(0,4))], Box.FRONT | Box.BACK | Box.LEFT | Box.RIGHT);
         blocks[horizontal][vertical].setTexture(tops[int(random(0,4))], Box.TOP);
         blocks[horizontal][vertical].drawMode(Shape3D.TEXTURE);

    }}}

void lightsAndCamera(){
 camera.feed();
 ambientLight(90,90,90);
 pushMatrix();
   translate(-totalLength/4, -300, +totalLength/4);
   noStroke();
   fill(255);
   shininess(1.0);
   sphere(20);
 popMatrix();
 lightFalloff(0.5, 0, 0.0);
 pointLight(250, 250, 255, -totalLength/4, -420, +totalLength/4); }

void draw(){
 lightsAndCamera();
 drawGround();
 for(int horizontal = 0; horizontal < blocksTotal; horizontal ++){
   for(int vertical = 0; vertical < blocksTotal; vertical ++){
         drawBlock(horizontal, vertical);
   }}
 drawSky();}

void drawBlock(int horizontal, int vertical){
  int xOffset = (blockLength + streetWidth) * horizontal - totalLength / 2;
  int zOffset = (blockLength + streetWidth) * vertical - totalLength / 2;
  pushMatrix();
    fill(200);
    stroke(255);
    translate(xOffset, 0, zOffset);
    blocks[horizontal][vertical].draw();
  popMatrix();}

void drawGround(){
  pushMatrix();
    fill(10);
    noStroke();
    box(totalLength, 0, totalLength);
  popMatrix();}

void drawSky(){
  fill(color(0,0,255));
  noStroke();
  sphere(totalLength*(1/sqrt(2)));}

void mouseMoved() {
  camera.circle(radians(mouseX - pmouseX));
  camera.dolly(mouseY - pmouseY);}

				Listing 2.1: Die Anwendung

				Links & Literatur

				
						http://processing.org/reference/texture_html

						http://www.lagers.org.uk/s3d4p/

						http://gdsstudios.com/processing/libraries/ocd/reference

						http://processing.org/reference/

				

			

		

	
		
			
				3	Multimedia mit Processing – Bilder und Töne

				In den vorherigen Kapiteln haben wir uns mit 2-D- und 3-D-Ausgaben im leeren Raum beschäftigt. Wir konnten uns zwar im Raum bewegen, aber das Ganze kann man auch interaktiver und lebhafter gestalten. Daher möchten wir in diesem Kapitel einen Augmented Reality Equalizer bauen: Wir zeichnen auf den ausgedruckten Marker-Codes im Bild der Webcam die Amplitude – also die Lautstärke – der einzelnen Frequenzbänder des Mikrofoneingangs.

				Mit der Bibliothek video ist es denkbar einfach, sich Bilder aus der Webcam zu besorgen. Der Einstiegspunkt ist die Klasse processing.video.Capture. Nachdem die Bibliothek über den Menüpunkt Sketch | Import Library | video eingebunden wurde, können wir Capture instanziieren. Es stehen verschiedene Konstruktoren zur Verfügung. In der Regel reicht schon new Capture(this) aus. Falls man von mehreren angeschlossenen Kameras eine bestimmte ansprechen oder die Auflösung selbst bestimmen möchte, stehen auch hierfür Konstruktoren bereit:

				
						available(): Gibt true zurück, falls mindestens eine Kamera gefunden wurde

						start(): Fängt mit der Aufnahme an

						stop(): Beendet die angefangene Aufnahme

						read(): Liest das aktuelle Bild in den Puffer, damit ist das Capture-Objekt auch als normales Bild zu verwenden – mehr später

						list(): Statische Methode, welche die vorhandenen Kameras und ihre Konfigurationen auflistet

				

				Toneingang

				Eine weitere spannende Bibliothek für Processing muss zunächst aus dem Internet geladen werden, was mit Processing 2.0.x sehr einfach ist. Analog zum letzten Kapitel rufen wir die Bibliothek über Sketch | Import Library | Add Library ... in der Suchmaske ab. In der Maske suchen wir nach minim und installieren die Bibliothek. 

				minim vereint drei Bibliotheken (JavaSoundAPI, Tritonus, MP3SPI) in sich und vereinfacht deren Umgang in typischer Processing-Manier: keine Callbacks, keine obskuren Arrays, keine Versionskonflikte. Mit minim lassen sich Sounddateien abspielen, Metadaten auswerten, Frequenzen analysieren und vieles mehr. In unserem Beispiel möchten wir einfach das Mikro abhören, um zum Beispiel auch eigene Töne zu analysieren.

				Analog zum Capture gibt es die Einstiegsklasse ddf.minim.Minim_, die es zu instanziieren gilt, z.B. mit new Minim(this). Zwei Methoden bieten einen sehr einfachen Zugriff auf unterschiedliche Tonquellen:

				
						loadFile(): Lädt eine Datei und stellt ein Player-Objekt für den Track zur Verfügung

						getLineIn(): Holt sich das Eingabesignal auf dem Eingabe-Port und stellt ein Input-Objekt zur Verfügung

				

				Mit getLineOut() lassen sich Töne und Effekte ausgeben, was aber in diesem Kapitel nicht weiter von Belang ist. Die Fast-Fourier-Transformation wiederum ist wichtig, um das Signal in verschiedene Frequenzbänder aufzulösen. Die zeitliche Länge und die Abtastrate des Signals muss beim Konstruktor mitgegeben werden. In der Regel kann dies von der Audioquelle ausgelesen werden: new FFT(source.bufferSize(), source.sampleRate()). 

				Augmented Reality

				Mikrofone, Töne, Kameras und Bilder sind bekannte Konzepte – Augmented Reality ist schon spannender: Wir werden in ein natürliches Bild vom Rechner generierte Grafiken einfügen.

				Hierzu muss der Rechner im eingehenden Bild nach so genannten Markern suchen. Das sind einfache Grafiken mit einem hohen Kontrast, die nicht rotationssymmetrisch sind. Aus der perspektivischen Verzerrung des Markers kann der Rechner die Lage des Markers rekonstruieren und einen berechneten 3-D-Körper auf dem Marker einrechnen. In unserem Beispiel sind die Körper einfache Kuben, was der Einfachheit des Kapitels und nicht den Einschränkungen der Anwendung geschuldet ist.

				Auch für Augmented-Reality-Anwendungen gibt es für Processing ein Framework: nyAR4psg. Es ist ein Wrapper für das nyARToolkit, das komplett in Java verfügbar ist. Die Bibliothek ist so einfach zu nutzen, dass man anhand weniger Codebeispiele die Funktionalität ableiten kann. Leider kann man es nicht anders erlernen, da die Dokumentation in Japanisch gehalten ist. 

				Bevor es ans Coden geht, müssen die Marker erstellt werden. Ein Marker besteht hier aus einer binären computerlesbaren Beschreibung als Teil der PATT-Datei. Dazu gibt es eine sichtbare Repräsentation der PPM-Grafik. Diese Kombinationen können mit dem genannten Kommandozeilentool [1] generiert werden. Die Grafiken werden anschließend ausgedruckt und ausgeschnitten. In unserem Beispiel werden wir mit vier verschiedenen Markern arbeiten.

				Mit den Markern zur Hand und auf der Festplatte binden wir die Bibliothek ein. Diesmal muss man die Bibliothek mit dem Browser herunterladen [2], das entpackte Verzeichnis in NyAR4psg umbenennen, da es Processing sonst nicht anerkennen wird, und in das Verzeichnis ~/Processing/libaries verschieben. Jetzt kann man es unter Sketch | Import Library ... einbinden. Die Klasse jp.nyatla.nyar4psg.MultiMarker dient als zentrales AR-Objekt und kann instanziiert werden. Hierzu benötigt man die Sketch, Höhe und Breite des Bilds, eine Kamerabeschreibung und ein Konfigurationsobjekt. 

				Da die optischen Systeme in den Kameras nicht perfekt sind (siehe auch das Bild unter [3]), müssen diese kalibriert werden, um möglichst genaue Angaben zu liefern. Das ist zeitaufwändig und für einfache Anwendungen nicht vonnöten, sodass man sich die Standarddatei camera_para.dat einfach aus dem Beispielverzeichnis der Bibliothek kopieren kann. Der Multimarkerinstanz kann man mit addMarker() Referenzen zu den PATT-Dateien nennen, damit das Framework im Webcambild nach den ausgedruckten Markern suchen kann. Dazu mehr später.

				[image: Abb3_1.png]

				Abbildung 3.1: Marker

				Set-up-Methode und Instanzenvariablen

				String myPatterns = "/Users/stefansiprell/Documents/Processing/ARStuff";

Minim minim;
MultiMarker nya;
int numMarkers = 4;

FFT fft;
Capture iSight;
AudioInput mic;

color[] colors = new color[numMarkers];
float[] frequencies = new float[numMarkers];

void setup(){
  size(640, 480, OPENGL);
  frameRate(30);
  textFont(createFont("Gill Sans", 80));
  textAlign(CENTER, CENTER);
  textSize(10);
   
  println("Spreche Kamera an.");
  iSight = new Capture(this, width, height, "FaceTime Camera (Built-in)");
  iSight.start();
  
  println("Spreche Micro an.");
  minim = new Minim(this);
  mic = minim.getLineIn();
  fft = new FFT(mic.bufferSize(), mic.sampleRate());

  println("Konfiguriere AR Framework.");
  nya = new MultiMarker(this, width, height, "camera_para.dat", NyAR4PsgConfig.CONFIG_DEFAULT);
  nya.setLostDelay(1);
  for(int i = 0; i <numMarkers; i++){
    nya.addARMarker("4x4_"+(i+1)+".patt", 80);
    colors[i] = color(random(255), random(255), random(255), 160); 
    frequencies[i] = ((float)20000/numMarkers)*(i+1);
  }

				Listing 3.1

				Die Konstrukte aus Listing 3.1 sollten mit der vorangegangenen Einführung einfach zu verstehen sein. TextFont ist an dieser Stelle neu und dient dazu, die Schriftart für folgende Schreiboperationen zu bestimmen. Das kann natürlich auch Courier, Arial oder ein beliebiger anderer Schriftsatz sein. Das Capture-Objekt wird mit dem Namen der Kamera parametrisiert, was aber nicht notwendig ist, wenn man nur eine Kamera angeschlossen hat. Das Mikrofon wird über dem Line-in-Port aufgemacht. Für die Konfiguration des ARToolkits ist ein Verweis auf die camera_part.data notwendig, die über Sketch | Add File ... zuerst eingebunden wird.

				In der Schleife werden die Marker geladen, und zwei Arrays werden bestückt. Im ersten Array werden zufällige halbtransparente Farben für die späteren Würfel definiert, und die Frequenzbänder werden im zweiten Array gespeichert. Je mehr Marker man verwendet, desto feingliedriger wird das Spektrum zwischen 0 und 20000 KHz aufgeteilt. Normalerweise sind 20 KHz das Ende bei digitalen Aufnahmen.

				void draw(){
  iSight.read();
  fft.forward(mic.mix);
  background(iSight);
  nya.detect(iSight);
  drawBoxes();
}

				Listing 3.2

				Die draw()-Methode in Listing 3.2 ist diesmal nicht nur für das Zeichnen des Bilds verantwortlich, sondern aktualisiert die Eingaben. Zunächst muss das nächste Bild aus der Kamera geladen werden. Das Bild wird zum einen als Hintergrund gesetzt, und zum anderen wird der Multimarker mit dem aktuellen Bild gefüttert. Die Marker werden in der letzten Operation gesucht und stehen weiter unten noch zur Verfügung. Hat der Fast-Fourier-Transformer das aktuell aufgenommene Signal analysiert, sind alle Daten vorhanden, um die Boxen zu zeichnen.

				void drawBoxes(){
  nya.setARPerspective();
  for(int i=0; i<numMarkers; i++){
    if((!nya.isExistMarker(i))){continue;}
    setMatrix(nya.getMarkerMatrix(i));
    drawBox(i);
  }
  perspective();
}

void drawBox(int i){
  //scale(1, -1);
  float scale = map(fft.getFreq(frequencies[i]), 0, 0.05, 0.7, 1.5);
  scale(scale);
  translate(0,0,20);
  lights();
  stroke(0);
  fill(colors[i]);
  box(40);
  noLights();
  translate(0,0,20.1);
  fill(0);
  text(""+frequencies[i], -20, -20, 40, 40);
}

				Listing 3.3 

				[image: Abb3_5.png]

				Abbildung 3.2: Fertige Ausgabe

				Zum Zeichnen der Kuben in die Ausgabe ruft die draw()-Methode die drawBoxes()-Methode auf. Die Perspektive für alle Ausgaben wird an dem Bild ausgerichtet. Anschließend gehen wir durch die Definitionen durch und lassen das Framework nach dem Marker auf dem Bild suchen. Wurde keines gefunden, suchen wir einfach weiter. Sonst setzen wir die Matrix für die folgenden Zeichenoperationen – damit ist sichergestellt, dass die 3-D-Körper immer auf dem Marker sitzen. Nachdem die Boxen gezeichnet worden sind, stellen wir noch die normale – nicht ausgerichtete – Perspektive wieder her.

				Das eigentliche Zeichnen der Box ist nicht neu (siehe auch vorheriges Kapitel). Wir stellen die Koordinaten auf den Kopf, damit Processing ein natürlicheres Koordinatensystem nutzt. Weiterhin definieren wir noch eine weitere Skalierung, um die Größe der Objekte zu bestimmen. Die Größe leitet sich von der Amplitude des jeweiligen Frequenzbands ab, das sich bei den normalen Mikrofon- und deren Digitalisierungseinstellungen zwischen 0 und 0,05 bewegt. Den Eingangswertebereich des Mikrofons bilden wir auf den Ausgabewertebereich zwischen 0,75 und 1,5 mit der Funktion map() ab. Mit den entsprechenden Licht- und Farbeinstellungen zeichnen wir die Würfel. Schlussendlich muss noch der Text für das jeweilige Frequenzband auf den Würfel gezeichnet werden. Für das weitere Vertiefen gibt es schöne Beispiele auf [4]. Zu guter Letzt sollte man sich auch die eigentliche ARToolkit [5] anschauen, von der sich alles ableitet. 

				Links & Literatur

				
						http://www.cs.utah.edu/gdc/projects/augmentedreality

						http://sourceforge.jp/projects/nyartoolkit/releases

						http://commons.wikimedia.org/wiki/File:Boesendorfer_003.jpg

						http://www.creativeapplications.net/processing/augmented-reality-with-processing-tutorial-processing

						http://www.hitl.washington.edu/artoolkit

				

			

		

	
		
			
				4	Professionelle Datenvisualisierung mit Java - Visualize me

				In den vorherigen Kapiteln des shortcuts haben wir die Programmierumgebung Processing vorgestellt und gezeigt, wie 2-D- und 3-D-Objekte im leeren Raum erstellt werden können, sowie die Realisierung von Augmented Reality mithilfe von Marker Codes dargelegt. Da Processing auf Java basiert, lässt es sich mit der ganzen Vielfalt der Java-Bibliotheken kombinieren und somit als Java-Entwickler einfacher in der jeweiligen IDE effektiver nutzen. In diesem vierten Kapitel des shortcut liegt der Fokus auf der Generierung von Grafiken.

				Die Datenmengen, mit denen wir es heute zu tun haben, werden immer größer. Infolgedessen wachsen auch die Informationen, die darin verborgen sind, ständig. Aufgrund der Masse an Daten und den darin enthaltenen Informationen, kann die Verarbeitung zu einer Herausforderung werden. Da wir Menschen Grafiken besser und leichter verstehen können als reine Datenansammlungen, ist Datenvisualisierung oft der einzige Weg, sich und anderen einen Überblick zu verschaffen. Infografiken ermöglichen es uns, auch komplexe Themen in kurzer Zeit umfassend zu verstehen. Dadurch, dass die Daten vereinfacht und in visueller Form dargestellt werden, wird das menschliche Gehirn unterstützt, da es grafische Zusammenhänge besser verarbeiten kann.

				Selbstverständlich wäre es zu unübersichtlich und auch nicht unbedingt sinnvoll, sämtliche gesammelten Daten zu visualisieren. Stattdessen geht es darum, die Daten grafisch darzustellen.

				Ben Fry, ein Genie der Datenvisualisierung und Mitentwickler von Processing, stellt in seinem Buch „Visualizing Data“ einen siebenstufigen Prozess vor, der den Weg der Antwortfindung sehr gut beschreibt.

				
						acquire (erwerben, beschaffen): Beschaffen der Daten aus Intranet, Internet, aus dem lokalen Laufwerk etc.

						parse (parsen): Strukturieren und kategorisieren der Daten.

						filter (filtern): Entfernen aller unnötigen Daten.

						mine (Muster erkennen): Anwenden von statistischen Techniken und solchen aus dem Gebiet des Data Minings, um Muster zu erkennen oder die Daten in einem mathematischen Kontext zu setzen.

						represent (repräsentieren): Wählen einer Basisdarstellung für die Repräsentation, wie beispielsweise Bar Graph, Liste, Baum etc.

						refine (verfeinern): Verfeinern der Basisdarstellung, um ein besseres Verständnis zu erzielen.

						interact (interagieren): Hinzufügen von Funktionen, um die Daten zu manipulieren, oder um zu kontrollieren, welche Features angezeigt werden sollen [1].

				

				Nach diesem siebenstufigen Prozess zu arbeiten, kann beeindruckende Ergebnisse hervorbringen. Processing bietet dabei Unterstützung für jeden einzelnen Schritt. Abbildung 4.1 veranschaulicht den Prozess noch einmal.

				[image: Abb4_1.png]

				Abbildung 4.1: Der siebenstufige Prozess nach Ben Fry

				An dieser Stelle möchte der Autor die Gelegenheit nutzen, nicht mit der PDE, sondern mit Eclipse zu entwickeln. Wenn man sich einmal an die Autovervollständigung gewöhnt hat, ist es schwierig, ohne sie zu arbeiten.

				Zunächst gilt es, ein einfaches Java-Projekt zu erstellen. Hierzu ist im Project Explorer mit der rechten Maustaste New | Project zu wählen und anschließend im Wizard Java Project. Über Wizard ist im nächsten Schritt der data-Ordner als Source-Ordner hinzuzufügen. Bei Bedarf kann dies auch erst im Nachhinein ohne Wizard getan werden. Sollte man zusätzliche Ressourcen, wie Bilder, Textfonts etc. benötigen, ist der data-Ordner gut geeignet, um sie dort abzulegen, da man sich dann um die Lokalisierung und das Laden der benötigten Daten nicht kümmern muss. Hier gilt das Prinzip „Convention over Configuration“ (CoC).

				Da in der Beschreibung in diesem Kapitel eine externe Bibliothek verwendet wird, ist der lib-Ordner anzulegen. Der Name ist dabei nicht zwingend, da hier keine CoC stattfindet. Die benötigten Libraries werden in den Klassenpfad aufgenommen und können in einem beliebigen Ordner liegen –sowohl in der fertigen JAR als auch außerhalb.

				Die Processing-Kernfunktionalität ist in der core.jar verpackt. Sie liegt in der Windows-Installation unter processing-2.0.1\core\library und muss in den Klassenpfad aufgenommen werden. Dafür ist sie in das \lib-Verzeichnis zu kopieren und im Klassenpfad aufzunehmen.

				Die Hauptklasse muss von PApplet erben. So bekommt man Zugriff auf alle schon bekannten Processing-Methoden und Variablen, wie fill(), noLoop(), keyPressed(), mousePressed(), width, height usw.

				Möchte man keine Multi-Applet-Anwendung bauen und doch seine eigenen Klassen zur grafischen Ausgabe schreiben, müssen diese mit einer Instanzvariablen vom Typ PApplet versehen und über den Konstruktor initialisiert werden. Eine detaillierte Ausführung folgt an späterer Stelle dieses Kapitels. Eine Schritt-für-Schritt-Anleitung ist daneben auch auf der Processing-Seite [2] zu finden.

				Über die Export-Funktion von Eclipse lässt sich das Projekt mit wenigen Klicks in eine lauffähige JAR exportieren. Hierzu ist für die Hauptklasse eine Main-Methode notwendig, wobei die Main-Methode von PApplet, je nach Wunsch, einen oder mehrere Parameter aufruft. Zuerst kommt der vollqualifizierende Klassenname, in dem die Main-Methode liegt, danach folgen die Aufrufargumente, verpackt in einem String-Array:

				public static void main(String args[]) {
PApplet.main("processing.graphic.tutorial.ForbesEu" , new String[] {"--full-screen"} );
}

				Laut Anleitung sollte man für diese Zwecke den present-Mode von Processing einschalten. Da dieser mittlerweile jedoch deprecated ist, empfiehlt es sich stattdessen full-screen zu verwenden. Welche Argumente hierbei zur Verfügung stehen, kann in der Javadoc [3] der PApplet-Klasse nachgelesen werden (Abb. 4.2).

				[image: Abb4_2.png]

				Abbildung 4.2: Die Projektstruktur

				Alternativ kann auch das Proclipsing-Plug-in [4] verwendet werden, auf das an dieser Stelle jedoch nicht näher eingegangen werden soll.

				Projektstruktur

				Nach der Vorbereitung der IDE für Processing, kann mit der Anwendung begonnen werden. In diesem Beispiel soll die Anzahl der Unternehmen pro EU-Staat in der Forbes-2000-Liste ausgewertet und dargestellt werden. Die vollständige Liste ist auf der Forbes-Website unter forbes.com einsehbar. In Listing 4.1 ist ein kurzer Ausschnitt zu sehen.

				Company;Industry;Country;Market Value;Sales;Profits;Assets;Rank;Forbes Webpage;
ICBC;Major Banks;China;237,3;134,8;37,8;2813,5;1;http://www.forbes.com;
China Construction Bank;Regional Banks;China;202;113,1;30,6;2241;2;http://www.forbes.com;
JPMorgan Chase;Major Banks;United States;191,4;108,2;21,3;2359,1;3;http://www.forbes.com;

				Listing 4.1

				public void setup() {
  size(750, 750);
  initFonts();
  acquire();
  parse();
  filterAndMine();
  createBarChart(euCountries);
  noLoop(); // einmal zeichnen reicht
}

				Listing 4.2

				In Listing 4.2 ist die bekannte setup()-Methode zu sehen, in der die Fenstergröße gesetzt wird. Danach werden die Fonts initialisiert sowie die Daten geholt und geparst. Die Fonts können mittels createFont() erstellt oder alternativ mit loadFont() vom data-Ordner geladen werden. Es ist möglich, in der PDE über Create Font… eigene Fonts zu erzeugen und sie im .vlw-Format zu speichern. Mit PFont.list() kann man sich darüber hinaus alle im System verfügbaren Fonts ausgeben lassen. Wichtig dabei ist, dass Operationen, die von der Festplatte oder aus dem Internet Daten laden, in der setup()-Methode platziert sind, damit die Daten nicht bei jedem Aufruf von draw() wieder geholt/geladen werden.

				Die Schritte

				Schritt 1: acquire

				Die Basisdaten können überall gespeichert sein – sowohl auf der Festplatte als auch im Intra- oder Internet. Alle Dateien, die über Processing geladen und/oder gespeichert werden, sind im UTF-8-Zeichensatz. Um an sie heranzukommen, können Processing-Boardmittel verwendet werden. Hierzu steht eine breite Palette an Möglichkeiten zur Verfügung. Mit loadTable() ist es möglich, CSV-Dateien in einem Table-Objekt zu laden. loadStrings() lädt den Dateiinhalt in einem String-Array, wobei in jedem String eine Zeile gespeichert wird. loadBytes() liefert die Datei in einem Byte-Array zurück. loadJSONArray() kann eine in [] eingeschlossene Liste von JSON-Objekten einlesen. loadXML() regelt das Einlesen von XML-Dateien. Tabelle 4.1 liefert einen Überblick über die Möglichkeiten des File Handlings.

				Tabelle 4.1: File Handling

				Damit die Applikation auch offline ihre Dienste tut, wurde im nächsten Schritt eine CSV-Datei im data-Ordner ausgewählt:

				acuire() {
  data = loadStrings(SOURCE_FILE);
}

				Schritt 2: parse

				Im Parse-Schritt gilt es, die Rohdaten in eine Struktur zu überführen, die geeignet für die weitere Verarbeitung ist. Dabei ist es notwendig, die gelieferten Daten zu inspizieren und zu verstehen, wie sie aufgebaut sind.

				Hierzu kann man selbstverständlich eine externe Library zum Einlesen von CSV verwenden. An dieser Stelle soll jedoch gezeigt werden, dass dies auch mit Processing einfach zu bewerkstelligen ist:

				void parse() {
...
int lineCount = data.length;
createTableHeader(data[0]);
for(int i = 1; i < lineCount; i++) {
forbesGlobal.addRow(replaceBtwColumns(data[i], 3, 8, ',', '.'));
}
...
}

				Die gewählte interne Struktur ist die einer processing.data.Table. Da Table nur kommaseparierte und Tab-separierte CSV-Formate unterstützt und die Daten im Beispiel zwischen Spalte drei und acht Double-Werte enthalten, die bei den Dezimalzeichen Kommata beinhalten, müssen diese angepasst werden. Eine Aufgabe, die gut zum Parse-Schritt passt.

				Es ist üblich, das Bereitstellen und Parsen der Daten nicht in Processing vorzunehmen. Stattdessen kann dies im Backend unter Verwendung mächtiger Bibliotheken geschehen.

				Zwar wirkt es, als gehöre das Parsen nicht unbedingt zum Prozess, jedoch gibt es gute Gründe, weshalb es dennoch ein fester Prozessbestandteil ist. Da die Daten in den meisten Fällen außerhalb der eigenen Kontrolle (man denke an sich ständig wechselnde Wetterdaten) und in nicht eigens definierten Formaten vorliegen, wird viel Zeit in die Interpretation und das Verstehen der Daten investiert.

				Schritt 3: filter und Schritt 4: mine

				In Schritt 3 gilt es, alle Daten zu entfernen, die für die Visualisierung nicht benötigt werden. In Schritt 4 werden die Daten anschließend in einen Kontext gesetzt, in dem sie besser zu handhaben sind (Listing 4.3).

				filterAndMine() {
...
  FloatDict euCountries = new FloatDict();
  while (rows.hasNext()) {
    TableRow row = rows.next();
    String country = row.getString(COUNTRY_INDEX);
    if (allEuCountries.contains(country)) {
      euCountries.add(country, 1f);
  }
  }
euCountries.sortValues();
...
}

				Listing 4.3

				Im Beispiel soll die Anzahl der Unternehmen pro EU-Land angezeigt werden. Dafür ist es notwendig, über alle Einträge zu iterieren und zu zählen, wie oft ein Unternehmen aus einem EU-Land gelistet ist.

				Für die Darstellung dieses Falls wurde ein Bar Chart gewählt. In diesem Zusammenhang bietet sich die FloatDict-Klasse von Processing an, um die gefilterten Daten aufzubewahren. Diese speichert ihre Werte, ähnlich einer Map, als (Key/Value-)Paare und verfügt über Methoden, sie einfach zu manipulieren. So ist es möglich, über set(key, amount) einen Wert zu setzen, über add(key, amount) einen Wert zu addieren usw. Dabei werden alle mathematischen Operationen unterstützt. Man kann sich alle Keys und Werte ausgeben lassen, sie sortieren und das Vorhandensein abfragen. Hierbei stehen ähnliche Strukturen, die Integers und Strings als Werte speichern, zur Verfügung. Die xyzList-Klassen (xyz -> [Float, Integer, String]) bieten eine ähnliche Funktionalität. Dabei wird jedoch nicht über einen Key auf den Wert zugegriffen, sondern über einen Index. Tabelle 4.2 bietet eine Übersicht der vorhandenen Strukturen.

				Tabelle 4.2: Strukturen (Namen sind selbsterklärend

				Schritt 5: represent

				Wie bereits erwähnt, wurde in diesem Beispiel ein Bar Chart ausgewählt. Selbstverständlich ist es möglich, alles selbst zu zeichnen und alle Abstände, Richtungen, x- und y-Achse, Überschriften und Werte zu berechnen oder alternativ eine externe Bibliothek zu nutzen. In diesem Fall fiel die Wahl auf giCenterUtil [5]. Diese muss aus dem Internet heruntergeladen, in den /lib-Ordner kopiert und in den Klassenpfad aufgenommen werden (Listing 4.4).

				createBarChart() {
  BarChart barChart new BarChart(this);
  barChart.setData(dict.valueArray());
  barChart.setBarLabels(dict.keyArray());
  barChart.setBarColour(color(200, 80, 80, 100));
  barChart.setBarGap(2);
  barChart.setValueFormat("###,###");
  barChart.showValueAxis(true);
  barChart.showCategoryAxis(true);
  barChart.draw(40, 40, width - 50, height - 50);
}

				Listing 4.4

				Damit der Bar Chart über die Funktionalität der PApplet verfügt, wird er damit erzeugt. Er bietet die Möglichkeit, die Werte und die Beschriftungen zu setzen. Hier zeigt die FloatDict-Klasse ihre Stärken, indem sie bereits die passenden Methoden bietet. Die Ländernamen werden auf der x-Achse und die Anzahl der Unternehmen auf der y-Achse dargestellt.

				Schritt 6: refine

				In diesem Schritt wird die Basisdarstellung angepasst, damit die Anzeige übersichtlicher und ansprechender wird. Betrachtet man die Darstellung, fällt schnell auf, dass die Namen sich überlappen, sodass an dieser Stelle ein Optimierungsbedarf besteht. Zwar ist zwischen den einzelnen Balken der Abstand vergrößerbar (setBarGap(int)), jedoch trägt dies zur Problemlösung nicht gänzlich bei (Abb. 4.3). Damit auch längere Namen, wie „Czech Republic“ oder „United Kingdom“ ausreichend Platz haben, ist der Graph sehr weit auseinanderzuziehen. Ebenso besteht die Möglichkeit, die Balken anstatt nach oben, zur Seite wachsen zu lassen. In diesem Fall würden die Ländernamen untereinander geschrieben, sodass sie sich nicht in die Quere kommen, was eine übersichtlichere Grafik zum Ergebnis hat. Dies wird mit transposeAxes(boolean) erreicht (Abb. 4.4):

				...
barChart.setBarGap(4);
barChart.transposeAxes(true);
...

				[image: Abb4_3.png]

				Abbildung 4.3: Basisdarstellung – Namen überlappen sich

				[image: Abb4_4.png]

				Abbildung 4.4: Die Grafik ist jetzt übersichtlicher

				Schritt 7: interact

				Interaktivität mit Processing umzusetzen, ist nicht kompliziert, da die Sprache sämtliche von Swing bekannten Aktionen, wie Mausklicks, Mausebewegungen, Tastenschläge, Eingabefelder etc., bietet. Beispielhaft wurden im nächsten Schritt zwei weitere Charts implementiert. Der erste beantwortet die Frage „Wie groß ist der Marktwert aller Unternehmen pro EU-Land?“, der zweite „Auf welchem Platz ist das bestplatzierte Unternehmen pro EU-Land?“. Zu den entsprechenden Ansichten gelangt man mit der rechten Maustaste bzw. mit der Taste r oder R. Die Default-Ansicht erhält man über die linke Maustaste.

				Fazit

				Processing ermöglicht es, mit relativ wenig Code einen umfangreichen Prozess zur Datenvisualisierung umzusetzen. Da Processing auf Java aufsetzt, kann ein Großteil der Verarbeitungsschritte auch im Backend durchgeführt werden und somit auch anspruchsvolle und professionelle Grafiken erstellt werden. Das gesamte beispielhafte Projekt kann auf GitHub [6] heruntergeladen werden. Enjoy!

				Links & Literatur

				
						Fry, Ben: Visualizing Data: Exploring and Explaining Data with the Processing Environment, 2008

						http://processing.org/tutorials/eclipse/

						http://processing.org/reference/javadoc/core/

						http://code.google.com/p/proclipsing/

						http://gicentre.org/utils/

						https://github.com/drobev/ProcessingGraphic

				Stefan Siprell (stefan.siprell@codecentric.de) ist Geschäftsführer der codecentric in Karlsruhe. Seine fachlichen und technischen Schwerpunkte liegen im Agile Software Enigneering. In der Freizeit beschäftigt er sich mit Arduino, Processing und Co.

				Dimitar Robev ist als IT-Berater im Java-Umfeld bei der EXXETA AG in Karlsruhe tätig. Seit ca. zehn Jahren entwickelt er Java-Anwendungen. In letzter Zeit ist er im Bankenumfeld unterwegs, wo er Real-Time-Analysesysteme entwickelt.

